Question: I know that it is possible to include AngularJs with Maven into a Spring project for instance but how would one include it with Gradle?

Looking into gradle repository I find no AngularJs entries. Maybe it's possible to fetch it from Maven repository with Gradle? But how would one do that.

Answer: 
Although it turned out in the discussion below the question that it doesn't make sense to download angular via gradle it can be done with the following piece of code:

```
repositories {
   ivy {
      name = 'AngularJS'
      url = 'https://code.angularjs.org/'
      layout 'pattern', {
         artifact '[revision]/[module](.[classifier]).[ext]'
      }
   }
}

configurations {
   angular
}

dependencies {
   angular group: 'angular', name: 'angular', version: '1.3.9', classifier: 'min', ext: 'js'
}

task fetch(type: Copy) {
   from configurations.angular
   into 'src/main/webapp/js'
   rename {
      'angular.js'
   }
}

```
Check out [webjars](http://www.webjars.org/), i highly recommend it.

I've just used it in Maven and Gradle project.  

Basically it is just what we needed, a vast collection of front-end projects and frameworks packaged in jars. 

* Add this dependency to your `build.gradle` script:

`compile 'org.webjars:angularjs:1.3.14'`
* Go to documentation section for a quick setup guide for Spring
* include `angular.js` or some other module from the jar in you script like a local resource 

`<script src="where-you-exposed-webjars/angular/1.3.14/angular.js"></script>`
* Or optionaly require it from `require.js` script